Q:

How to specify cell when copying cell value?

I'm trying to open a few workbooks in a folder to pull some data to the ThisWorkbook. I'm having problems with Cells notation.
Option Explicit

Sub UpdateAttendance()
    GetData
End Sub

Private Sub GetData()
    Dim wb As Workbook, ws As Worksheet
    Dim fso As Object
    Dim folder As Object
    Dim wbFile As Object
    Set fso = CreateObject("Scripting.FileSystemObject")
    Set folder = fso.GetFolder("C:\Users\8865\Documents\excel\relationships\registers\regs")
    Dim y As Long

    y = ThisWorkbook.Sheets(1).Cells(Rows.Count, 1).End(xlUp).Row + 1

    For Each wbFile In folder.Files
        If fso.GetExtensionName(wbFile.Name) = "xlsx" Then
        Set wb = Workbooks.Open(wbFile.Path)
        For Each ws In wb.Sheets
            'ThisWorkbook.Sheets(1).Range(Cells(1, 1)).Value = ws.Range("E19").Value --> Doesn't Work
            'ThisWorkbook.Sheets(1).Range(Cells(1, 2)).Value = ws.Range("E35").Value --> Doesn't work
            ThisWorkbook.Sheets(1).Range("A1").Value = ws.Range("E35").Value ' --> Works fine but I need to loop it to change the row so I have to use Cells
            ThisWorkbook.Sheets(1).Range("B1").Value = ws.Range("E19").Value ' --> works fine but I need to loop it to change the row so I have to use Cells
            'ThisWorkbook.Sheets(1).Range(Cells(1, 1), Cells(1, 3)).Value = ws.Range("E19, E35, E40").Value --> Is it possible to do something like that?

            y = y + 1
        Next ws

        wb.Close
        End If

    Next wbFile

End Sub

See above, I put some comments at the end of the line to indicate what works and what doesn't.
EDIT:
Just to clarify, I'd do the following about iterating but it throws an error

Runtime error 1004, Application defined or object defined error)

y = ThisWorkbook.Sheets(1).Cells(Rows.Count, 1).End(xlUp).Row + 1

For Each wbFile In folder.Files
    If fso.GetExtensionName(wbFile.Name) = "xlsx" Then
        Set wb = Workbooks.Open(wbFile.Path)
        For Each ws In wb.Sheets

            ThisWorkbook.Sheets(1).Range(Cells(y, 1)).Value = ws.Range("E19").Value
            ThisWorkbook.Sheets(1).Range(Cells(y, 2)).Value = ws.Range("E35").Value
            y = y + 1
        Next ws

        wb.Close
    End If

Next wbFile

A:

In the second coding example, replace:
ThisWorkbook.Sheets(1).Range(Cells(y, 1)).Value = ws.Range("E19").Value

with:
ThisWorkbook.Sheets(1).Cells(y, 1).Value = ws.Range("E19").Value

You don't need Cells() within Range()

